

Solving the TTC Train Benchmark Case with FunnyQT

Tassilo Horn

Institute for Software Technology, University Koblenz-Landau, Germany

horn@uni-koblenz.de

This paper describes the FunnyQT solution to the TTC 2015 Train Benchmark transformation case. The solution solves all core and all extension tasks. FunnyQT is a model querying and model transformation library for the functional Lisp-dialect Clojure providing a comprehensive and efficient querying and transformation API, many parts of which are provided as task-oriented embedded DSLs.

1 Introduction

This paper describes the FunnyQT¹ [2, 3] solution of the TTC 2015 Train Benchmark Case [5]. All core and extension tasks have been solved. The solution project is available on Github², and it is set up for easy reproduction on a SHARE image³.

FunnyQT is a model querying and transformation library for the functional Lisp dialect Clojure⁴. Queries and transformations are plain Clojure programs using the features provided by the FunnyQT API.

As a Lisp, Clojure provides strong metaprogramming capabilities that are exploited by FunnyQT in order to define several *embedded domain-specific languages* (DSL, [1]) for different querying and transformation tasks.

FunnyQT is designed with extensibility in mind. By default, it supports EMF [4] models and JGraLab⁵ TGraph models. Support for other modeling frameworks can be added without having to touch FunnyQT's internals.

The FunnyQT API is structured into the following namespaces, each namespace providing constructs supporting concrete querying and transformation use-cases:

funnyqt.emf EMF-specific model management API

funnyqt.tg JGraLab/TGraph-specific model management API

funnyqt.generic Protocol-based, generic model management API

funnyqt.query Generic querying constructs such as quantified expressions or regular path expressions

funnyqt.polyfns Constructs for defining polymorphic functions dispatching on metamodel types

funnyqt.pmatch Pattern matching constructs

funnyqt.relational Constructs for logic-based, relational model querying inspired by Prolog

funnyqt.in-place In-place transformation rule definition constructs

funnyqt.model2model Out-place transformation definition constructs similar to ATL or QVT Operational Mappings

funnyqt.extensional Transformation API similar to GReTL

¹<http://funnyqt.org>

²<https://github.com/tsdh/ttc15-train-benchmark-funnyqt>

³The SHARE image name is ArchLinux64_TTC15-FunnyQT

⁴<http://clojure.org>

⁵<http://jgralab.github.io>

funnyqt.bidi Constructs for defining bidirectional transformations similar to QVT Relations

funnyqt.coevo Constructs for transformations that evolve a metamodel and a conforming model simultaneously at runtime

funnyqt.visualization Model visualization

funnyqt.xmltg Constructs for querying and modifying XML files as models conforming to a DOM-like metamodel

For solving the train benchmark case, the *funnyqt.emf* and *funnyqt.in-place* namespaces have been used.

2 Solution Description

In this section, the individual tasks are discussed one by one. They are all implemented as in-place transformation rules supported by FunnyQT's *funnyqt.in-place* transformation DSL. The rules' repair actions simply call the CRUD functions of the EMF-specific *funnyqt.emf* namespace.

Task 1: PosLength. The transformation rule realizing the *PosLength* task is given below.

```
1 (defrule ~:forall ~:recheck pos-length [g]
2   [segment<Segment>
3     :when (<= (aget-raw segment :length) 0)]
4   (eset! segment :length (inc (- (aget-raw segment :length))))
```

The `defrule` macro defines a new in-place transformation rule with the given name (`pos-length`), formal parameters (`[g]`), a pattern (`[segment<Segment>...]`), and one or many actions to be applied to the pattern's matches (`(eset! ...)`). The first formal parameter must denote the model the rule is applied to, so here the argument `g` denotes the train model when the rule is applied using (`pos-length my-train-model`).

The pattern matches a node called `segment` of metamodel class `Segment`. Additionally, the segment's length must be less or equal to zero. The action says that the segment's `length` attribute should be set to the incremented negation of the current length.

The normal semantics of applying a rule is to find one single match of the rule's pattern and then execute the rule's actions on the matched elements. The modifier `~:forall` changes this behavior to finding all matches first, and then applying the actions to each match one after the other. FunnyQT automatically parallelizes the pattern matching process of such forall-rules under certain circumstances like the JVM having more than one CPU available and the pattern declaring at least two elements to be matched.

The `~:recheck` modifier causes the rule to recheck if a pre-calculated match is still conforming the pattern just before executing the rule's actions on it. This can be needed for forall-rules whose actions possibly invalidate matches of the same rule's pattern, e.g., when the application of the action to a match m_i cause another match m_j to be no valid match any longer.

Task 2: SwitchSensor. The transformation rule realizing the *SwitchSensor* task is given below.

```
5 (defrule ~:forall ~:recheck switch-sensor [g]
6   [sw<Switch> -!<:sensor>-> <>]
7   (eset! sw :sensor (ecreate! nil 'Sensor)))
```

It matches a switch `sw` which is not contained by some sensor. The exclamation mark of the edge symbol `-!<:sensor>->` specifies that no such reference must exist, i.e., it specifies a negative application condition. The action fixes this problem by creating a new `Sensor` and assigning that to the switch `sw`.

Task 3: SwitchSet. The `switch-set` rule realizes the *SwitchSet* task. Its definition is given below.

```

8 (def Signal-GO (enum-literal 'Signal.GO))
9 (defrule ^:forall ^:recheck switch-set [g]
10 [route<Route> -<:entry>-> semaphore
11  :when (= (eget-row semaphore :signal) Signal-GO)
12  route -<:follows>-> swp -<:switch>-> sw
13  :let [swp-pos (eget-row swp :position)]
14  :when (not= (eget-row sw :currentPosition) swp-pos)]
15  (eset! sw :currentPosition swp-pos))

```

It matches a `route` with its entry `semaphore` where the semaphore's signal is `Signal.GO`. The route follows some switch position `swp` whose switch's current position is different from that of the switch position. The fix is to set the switch's current position to the position of the switch position `swp`.

Note that there are no metamodel types specified for the elements `semaphore`, `swp`, and `sw` because those are already defined implicitly by the references leading to them, e.g., all elements referenced by a route's `follows` reference can only be instances of `SwitchPosition` according to the metamodel. Thus, FunnyQT doesn't require the transformation writer to encode tautologies in her patterns⁶.

Extension Task 1: RouteSensor. The extension task *RouteSensor* is realized by the `route-sensor` rule given below.

```

16 (defrule ^:forall ^:recheck route-sensor [g]
17 [route<Route> -<:follows>-> swp -<:switch>-> sw
18  -<:sensor>-> sensor --!<> route]
19  (eadd! route :definedBy sensor))

```

It matches a `route` that follows some switch position `swp` whose switch `sw`'s `sensor` is not contained by the `route`. The repair action is to assign the `sensor` to the `route`.

Extension Task 2: SemaphoreNeighbor. The second and last extension task *SemaphoreNeighbor* is realized by the `semaphore-neighbor` rule defined as shown below.

```

20 (defrule ^:forall ^:recheck semaphore-neighbor [g]
21 [route1<Route> -<:exit>-> semaphore
22  route1 -<:definedBy>-> sensor1 -<:elements>-> te1
23  -<:connectsTo>-> te2 -<:sensor>-> sensor2
24  --<> route2<Route> -!<:entry>-> semaphore
25  :when (not= route1 route2)]
26  (eset! route2 :entry semaphore))

```

It matches a route `route1` which has an exit `semaphore`. Additionally, `route1` is defined by a sensor `sensor1` which contains some track element `te1` that connects to some track element `te2` whose sensor is `sensor2`. This `sensor2` is contained by some other route `route2` which does not have `semaphore` as entry semaphore. The fix is to set `route2`'s entry reference to `semaphore`.

2.1 Deferred Rule Actions

As already mentioned above, the normal semantics of a forall-rule is to compute all matches of the rule's pattern first (possibly in parallel), and then apply the rule's actions on every match one after the other. However, the case description strictly separates the computation of matches from the repair transformations.

⁶In fact, if there are types specified, those will be checked. So omitting them when they are not needed also results in slightly faster patterns.

FunnyQT also provides stand-alone patterns. Using them, one could have defined patterns for finding occurrences of the five problematic situations in a train model, and separate functions for the repair actions where the latter receive one match of the corresponding pattern and fix that.

But for in-place transformation rules, FunnyQT also provides *rule application modifiers*. Concretely, any in-place transformation rule `r` can be called as `(as-pattern (r model))` in which case it behaves as a pattern. That is, where a normal rule would usually find one match and apply its actions on it and a forall-rule would usually find all matches and apply its actions to each of them, when called with `as-pattern`, a rule simply returns the sequence of its matches. With a normal rule, this sequence is a lazy sequence, i.e., the matches are not computed until they are consumed. With a forall-rule, the sequence is fully realized, i.e., all matches are already pre-calculated (possibly in parallel).

The second FunnyQT rule application modifier is `as-test`, and this is what is highly suitable for this transformation case. When a rule `r` is applied using `(as-test (r model))`, it behaves almost as without modifier but instead of applying the rule's actions immediately, it returns a closure of arity zero (a so-called *thunk*) which captures the rule's match and the rule's actions. Invoking the thunk causes the actions to be applied on the match. Thus, the caller of the rule gets the information if the rule was applicable at all, and if it was applicable, he can decide if he wants to apply it or not. And when he applies it, the pattern matching part is already finished and only the actions are applied on the pre-calculated match the thunk closes over. The following snippet illustrates the behavior.

```
(if-let [thunk (as-test (r model))]
  (if (< (Math/random) 0.5)
    (thunk) ;; The rule was applicable, and here its actions are executed
    (println "The coin toss decided to skip the application of the rule's matches"))
  (println "The rule is not applicable"))
```

One interesting point is that the thunk returned by calling a rule as a test has some metadata attached⁷. For this TTC case, only the `:match` metadata entry is important. As its name suggests, its value is the match of the rule's pattern on which the thunk applies the actions.

In case of a forall-rule `r`, `(as-test (r model))` doesn't return a single thunk but a vector of thunks, one thunk per match of the rule's pattern. This is exactly what is needed for solving this transformation case.

Finally, a function is defined that receives a rule `r` and a train model `g` and executes the rule as a test.

```
27 (defn call-rule-as-test [r g]
28   (as-test (r g)))
```

This is only needed in order to be able to call the rules as tests from Java. The reason is that only functions (and rules which are functions, too) can be referred to from Java but `as-test` is a macro which does its magic at compile-time.

These 28 lines of Clojure code form the complete functional part of the FunnyQT solution that solves all core and extension tasks. The comparator used for sorting the matches in the benchmark framework is also implemented using Clojure/FunnyQT and discussed in section 2.2. Additionally, there is a plain-Java glue project which implements the interfaces required by the benchmark framework and simply delegates to the Clojure/FunnyQT part of the solution. This glue project is briefly discussed in section 2.3 on the next page.

2.2 Match Comparison

The case description demands that solutions provide comparators that are to be used for sorting matches. All comparators simply compare two matches element by element with respect to the values of the id-

⁷Almost any Clojure object (functions, symbols, vars, collections, etc.) can have metadata attached. Metadata doesn't affect equality, i.e., two Clojure objects that differ only in their metadata are still equal.

attribute. The order in which the elements of two matches have to be compared is defined separately for each task.

Instead of providing one comparator per task, the FunnyQT solution provides one higher-order function `make-match-comparator` which receives the names of the match elements to be compared and then returns a suitable comparator. The function's definition is given below.

```

1 (defn make-match-comparator [& kws]
2   (fn [t1 t2]
3     (loop [kws kws]
4       (if (seq kws)
5         (let [m1 ((first kws) (:match (meta t1)))
6               m2 ((first kws) (:match (meta t2)))]
7           (let [r (compare (aget m1 :id) (aget m2 :id))]
8             (if (zero? r)
9               (recur (rest kws))
10              r)))
11         0))))

```

The `make-match-comparator` function is to be used as follows.

```

(make-match-comparator :segment)
;=> comparator for pos-length
(make-match-comparator :sw)
;=> comparator for switch-sensor
(make-match-comparator :semaphore :route :swp :sw)
;=> comparator for switch-set
(make-match-comparator :route :sensor :swp :sw)
;=> comparator for route-sensor
(make-match-comparator :semaphore :route1 :route2 :sensor1 :sensor2 :te1 :te2)
;=> comparator for semaphore-neighbor

```

`make-match-comparator` returns a function of two arguments. In Clojure, every function implements the `java.util.Comparator` interface and thus any function of two arguments which returns an integer can be used as such.

The returned comparator function receives two thunks `t1` and `t2` that were returned by calling any of the five above rules as tests. Then it iterates the keywords denoting the matched element names. Lines 5 and 6 extract the actual elements of the thunks' matches which are denoted by the first keyword. Then line 7 compares the id attribute values. If the comparison returns a non-zero value and thus these elements are different, the value is simply returned. Otherwise, the remaining keywords are tested one after the other. Only if no distinction between matches can be made after considering all given keywords, zero is returned.

2.3 Gluing the Solution with the Framework

Typically, open-source Clojure libraries and programs are distributed as JAR files that contain the source files rather than byte-compiled class files. This solution does the same, and that JAR is deployed to a local Maven repository from which the Maven build infrastructure of the benchmark framework can pick it up.

Then, in the FunnyQT glue project the rules and functions from above are referred to like shown in the next listing.

```

1 private final static String SOLUTION_NS = "ttc15-train-benchmark-funnyqt.core";
2 Clojure.var("clojure.core", "require").invoke(Clojure.read(SOLUTION_NS));
3 final static IFn POS_LENGTH = Clojure.var(SOLUTION_NS, "pos-length");
4 ...
5 final static IFn CALL_RULE_AS_TEST = Clojure.var(SOLUTION_NS, "call-rule-as-test");

```

In line 2, the solution namespace `ttc15-train-benchmark-funnyqt.core` is required⁸. The `Clojure`

⁸`require` is kind of Clojure's equivalent to Java's `import` statement.

class provides a minimal API for loading Clojure code from Java. When requiring a namespace as above, it will be parsed and compiled to JVM byte-code just in time⁹.

Thereafter, the solution’s in-place transformation rules and the `call-rule-as-test` function are referred to. `IFn` is a Clojure interface whose instances are Clojure functions that can be called using the `invoke()` method as can be seen in the definition of the glue project’s `BenchmarkCase.check()` method shown below.

```

1 @Override
2 protected final Collection<Object> check() throws IOException {
3     matches = (Collection<Object>) FunnyQTBenchmarkLogic.CALL_RULE_AS_TEST
4         .invoke(rule, this.resource);
5     // If the rule has no matches it returns nil/null but the framework
6     // wants a Collection.
7     if (matches == null) {
8         matches = new LinkedList<Object>();
9     }
10    return matches;
11 }

```

In that code, `rule` is one of the rule `IFns` `POS_LENGTH`, `SWITCH_SET`, et cetera, and they are called via the `call-rule-as-test` function to make them return one think per match instead of performing the rules’ repair actions immediately.

The implementation of the `BenchmarkCase.modify()` method is even simpler.

```

1 @Override
2 protected final void modify(Collection<Object> matches) {
3     for (Object m : matches) {
4         ((IFn) m).invoke();
5     }
6 }

```

Since the rules are called as tests and thus return thinks that apply the rule’s actions, those simply need to be invoked, and that’s it.

3 Evaluation

Correctness and Completeness. The FunnyQT solution implements all core and all extension tasks exactly as demanded by the case description, thus it is complete. When run in the benchmark framework, all assertion it checks are satisfied, thus the solution is also correct.

According to the case description, this gives a *correctness and completeness score* of 15 points.

Conciseness. The FunnyQT solution consists of 28 NCLOC of FunnyQT/Clojure code for the five rules with their patterns and repair actions, and the function `call-rule-as-test`.

In comparison, the reference EMF-IncQuery solution has about 70 NCLOC only for the patterns plus 69 NCLOC of plain Java code for the repair actions. Honestly, more than half of the lines of the Java repair actions are boilerplate code. So by a fair measure one could say that the EMF-IncQuery solution consists of a little bit less than 100 lines of code which actually implement the solution and are no boilerplate.

The reference Java solution consists of 305 NCLOC for the queries and repair actions (excluding the classes for the custom match representations). On the one hand, one could subtract maybe 10% for boilerplate code but on the other hand, when encoding queries and transformations in Java, boilerplate code is a fact you have to live with.

⁹If the Clojure code was distributed in a pre-compiled form, the resulting classes would simply be loaded.

To sum up, the FunnyQT solution is about three times shorter than the EMF-IncQuery solution and about ten times shorter than the Java solution. Thus, if there is no even more concise solution, it deserves a *conciseness score* of 15 points.

Readability. Readability is a very subjective matter, of course, so I won't suggest a *readability score* value here. However, there are some strong points with respect to readability.

- The queries (patterns) and repair actions are bundled in one concise in-place transformation rule each. In contrast, in the EMF-IncQuery as well as the Java solution the queries and repair transformations are strictly separated. But the latter still depend on the former in that they can only work with the matches produced by the former.
- FunnyQT's pattern matching DSL used to specify the rules' patterns is both concise and readable. It should be easy to understand for graph transformation experts, especially if they have used other textual graph transformation languages such as *GrGen.NET* before. It should also be easy to understand for any Clojure programmer because it strictly conforms to the style guidelines and best practices there, too.

Performance. The figures 1 on the following page and 2 on page 9 show the runtimes of the FunnyQT, EMF-IncQuery, and Java solutions for all models up to the size of 4096 (6,221,501 elements, 11,460,009 references) and the fixed and proportional strategies when being run by the train benchmark framework. The measurements have been performed on an 2.7 GHz 8-core machine with 16 GB of RAM being dedicated to the JVM process.

The FunnyQT and Java solutions have very similar performance characteristics. With the larger models, the *load & check* phase is dominated by the load time for the EMF models. The EMF-IncQuery is almost three times slower in this phase because the initialization of its incremental pattern matching engine needs quite some time. However, performing the *repair & recheck* ten times takes at most one and a half second, so here the incremental approach really pays off.

In contrast, the FunnyQT solution takes 5-6 seconds per *repair & recheck* iteration, and the Java solution takes about 3 seconds per iteration. Both don't perform incremental matching but each *check* requires a completely new re-computation of all matches.

It is a bit astonishing that the Java solution is faster than the FunnyQT solution. The Java solution uses an ideal search algorithm for the queries. The functions the FunnyQT rules expand to use a very similar algorithm, and at least for the patterns matching more than two elements, the search is performed in parallel exploiting all of the test system's cores. Of course, parallelization also involves some overhead, and the code generated from the FunnyQT rules is not as optimal as hand-written code.

One big difference between the FunnyQT and Java solutions is that the former rechecks matches before applying the repair actions as required by the case description. The Java solution doesn't do this costly action and thus can be considered not completely correct. That's another big benefit of an incremental approach like EMF-IncQuery: the matches can never be invalid so there is no point in rechecking.

One other thing to note is that the FunnyQT solution runs quite a bit faster when being run in its own project instead of being invoked by the benchmark framework. In the former case, its performance (with rechecking of matches) is on par with the performance of the Java solution (without rechecking of matches). I suspect that the performance loss in the benchmark framework is due to the fact that the benchmark framework executes each test in a separate, newly spawned JVM instance without any warmup phase. In addition, the FunnyQT solution is loaded and compiled dynamically from its sources.




Figure 1: Results of the performance measurements (*fixed* strategy)

Both reasons might result in less optimizations being performed by the JVM HotSpot compiler, and even if they are performed, they contribute to the overall measured runtime.

In summary, the ranking between the three solutions when adding up the measured runtimes for both the *load & check* phase and ten iterations of the *repair & recheck* phase is Java first, then FunnyQT, and then EMF-IncQuery. However, if one increases the iteration count for the *repair & recheck* phase, the EMF-IncQuery solution will outperform the other two solutions pretty easily simply because this scenario matches exactly what an incremental pattern matching approach is meant for.

4 Conclusion

This paper described the FunnyQT solution to the TTC 2015 Train Benchmark case. It solves all three core tasks and all two extension tasks, and the benchmark framework provides evidence that it delivers correct results.

The solution is extremely concise. Its five rules and one function amount to a total of only 28 lines of code (excluding comments and empty lines).

With respect to readability, FunnyQT’s pattern matching/in-place transformation rule DSL should be quite familiar to both people with a graph transformation background and people with a Clojure background. A strong point is that the patterns matching invalid subgraphs and the corresponding repair




Figure 2: Results of the performance measurements (*proportional* strategy)

actions are defined in one place as transformation rules. Nevertheless, the matching part and the application of the repair actions on (only parts of) the matches is well-supported using FunnyQT’s rule application modifier macro `as-test`.

The FunnyQT solution also performs very well for large models although the incremental pattern matching approach inherent to the EMF-IncQuery solution can easily outperform the FunnyQT solution when increasing the iteration count of the *repair & recheck* phase from 10 iterations to 20 or more. Like with all traditional, search-based pattern matching approaches, FunnyQT’s prime scenario isn’t the incremental one. If one would measure the time needed for finding all invalid subgraphs and repairing them all at once, then FunnyQT is a very competitive approach due to its feature of performing parallel pattern matching automatically for forall-rules on multi-core machines. In scenarios where rules should find just one match and perform actions on that, FunnyQT’s normal rules that perform lazy pattern matching also achieve a high performance.

Another aspect speaking in favor of FunnyQT is its comprehensiveness. It provides APIs and embedded DSLs for almost any querying and transformation task conceivable. Since the FunnyQT project has been started in 2012, there hasn’t been a single TTC case for which no FunnyQT solution has been provided.

References

- [1] Martin Fowler (2010): *Domain-Specific Languages*. Addison-Wesley Professional.
- [2] Tassilo Horn (2013): *Model Querying with FunnyQT - (Extended Abstract)*. In Keith Duddy & Gerti Kappel, editors: *ICMT, Lecture Notes in Computer Science 7909*, Springer, pp. 56–57.
- [3] Tassilo Horn (2015): *Graph Pattern Matching as an Embedded Clojure DSL*. In: *International Conference on Graph Transformation - 8th International Conference, ICGT 2015, L'Aquila, Italy, July 2015*. To appear.
- [4] Dave Steinberg, Frank Budinsky, Marcelo Paternostro & Ed Merks (2008): *EMF: Eclipse Modeling Framework*, 2 edition. Addison-Wesley Professional.
- [5] Gábor Szárnyas, Oszkár Semeráth, István Ráth & Dániel Varró (2015): *The TTC 2015 Train Benchmark Case for Incremental Model Validation**. In: *Transformation Tool Contest 2015*.